The Jaro-Winkler distance is a metric for measuring the edit distance between words.
It is similar to the more basic Levenstein distance but the Jaro distance also accounts
for transpositions between letters in the words. With the Winkler modification to the Jaro
metric, the Jaro-Winkler distance also adds an increase in similarity for words which
start with the same letters (prefix).

The Jaro-Winkler distance is a modification of the Jaro similarity metric, which
measures the similarity between two strings. The Jaro similarity is 1.0 when strings are
identical and 0 when strings have no letters in common. Distance measures such as the
Jaro distance or Jaro-Winkler distance, on the other hand, are 0 when strings are
identical and 1 when they have no letters in common.

The Jaro similarity between two strings s1 and s2, simj, is defined as

 simj = 0     if m is 0.
 simj = ( (m / length of s1) + (m / length of s2) + (m - t) / m ) / 3     otherwise.
Where:

 



m


{\displaystyle m}

   is the number of matching characters (the same character within max(|s1|, |s2|)/2 - 1 of one another);
 



t


{\displaystyle t}

   is half the number of transpositions (a shared character placed in different positions).


The Winkler modification to Jaro is to check for identical prefixes of the strings.

If we define the number of initial (prefix) characters in common as:

l = the length of a common prefix between strings, up to 4 characters

and, additionally, select a multiplier (Winkler suggested 0.1) for the relative importance of the
prefix for the word similarity:

 p    =   0.1

The Jaro-Winkler similarity can then be defined as

simw = simj + lp(1 - simj)

Where:

 simj   is the Jaro similarity.
 l   is the number of matching characters at the beginning of the strings, up to 4.
 p   is a factor to modify the amount to which the prefix similarity affects the metric.
Winkler suggested this be 0.1.



The Jaro-Winkler distance between strings, which is 0.0 for identical strings, is then defined as

dw = 1 - simw

String metrics such as Jaro-Winkler distance are useful in applications such as spelling checkers,
because letter transpositions are common typing errors and humans tend to misspell the middle portions
of words more often than their beginnings. This may help a spelling checker program to generate
better alternatives for misspelled word replacement.

The task

Using a dictionary of your choice and the following list of 9 commonly misspelled words:

"accomodate", "definately", "goverment​", "occured", "publically", "recieve​", "seperate", "untill", "wich​"

 Calculate the Jaro-Winkler distance between the misspelled word and words in the dictionary.
 Use this distance to list close alternatives (at least two per word) to the misspelled words.
 Show the calculated distances between the misspelled words and their potential replacements.
 See also
   Wikipedia page: Jaro–Winkler distance.
   Comparing string similarity algorithms. Comparison of algorithms on Medium


# download("http://users.cs.duke.edu/~ola/ap/linuxwords", "linuxwords.txt")
const words = read("linuxwords.txt", String) |> split .|> strip
 
function jarowinklerdistance(s1, s2)
    if length(s1) < length(s2)
        s1, s2 = s2, s1
    end
    len1, len2 = length(s1), length(s2)
    len2 == 0 && return 0.0
    delta = max(0, len2 ÷ 2 - 1)
    flag = zeros(Bool, len2)  # flags for possible transpositions, begin as false
    ch1_match = eltype(s1)[]
    for (i, ch1) in enumerate(s1)
        for (j, ch2) in enumerate(s2)
            if (j <= i + delta) && (j >= i - delta) && (ch1 == ch2) && !flag[j]
                flag[j] = true
                push!(ch1_match, ch1)
                break
            end
        end
    end
    matches = length(ch1_match)
    matches == 0 && return 1.0
    transpositions, i = 0, 0
    for (j, ch2) in enumerate(s2)
        if flag[j]
            i += 1
            transpositions += (ch2 != ch1_match[i])
        end
    end
    jaro = (matches / len1 + matches / len2 + (matches - transpositions/2) / matches) / 3.0
    commonprefix = count(i -> s1[i] == s2[i], 1:min(len2, 4))
    return 1 - (jaro + commonprefix * 0.1 * (1 - jaro))
end
 
function closewords(s, maxdistance, maxtoreturn)
    jw = 0.0
    arr = [(w, jw) for w in words if (jw = jarowinklerdistance(s, w)) <= maxdistance]
    sort!(arr, lt=(x, y) -> x[2] < y[2])
    return length(arr) <= maxtoreturn ? arr : arr[1:maxtoreturn]
end
 
for s in ["accomodate", "definately", "goverment", "occured", "publically",
    "recieve", "seperate", "untill", "wich"]
    println("\nClose dictionary words ( distance < 0.15 using Jaro-Winkler distance) to '$s' are:")
    println("    Word      |  Distance")
    for (w, jw) in closewords(s, 0.15, 5)
        println(rpad(w, 14), "| ", Float16(jw))
    end
end
 